Maple-assisted proof of formula for A296015

Robert Israel

3 December 2017

There are $2^8 = 256$ possible configurations for a 2×4 sub-array. Consider the 256×256 transition matrix T such that $T_{ij} = 1$ if the bottom two rows of a 3×4 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the middle row is compatible with both i and j, and each 1 in that row is king-move adjacent to 1 or 4 1's), and 0 otherwise. The following Maple code computes it. I'm encoding a configuration

$$\begin{bmatrix} b_1 & b_2 & b_3 & b_4 \\ b_5 & b_6 & b_7 & b_8 \end{bmatrix}$$

as $b+1$ where $b_1 b_2 b_3 b_4 b_5 b_6 b_7 b_8$ is the binary representation of b. The $+1$ is needed because matrix indices start at 1 rather than 0.

```
> q:= proc(a,b) local r,s,t,M,i;
    s:= floor((a-1)/16);
    if s <> (b-1) mod 16 then return 0 fi;
    s:= convert(s+16,base,2);
    r:= convert(16+floor((b-1)/16),base,2);
    t:= convert(16+ ((a-1) mod 16),base,2);
    M:= Vector(4);
    for i from 1 to 3 do if s[i] = 1 and s[i+1] = 1 then M[i]:= M
 [i]+1; M[i+1]:= 1 fi od;
    for i from 1 to 4 do if s[i]=1 then
      M[i]:= M[i]+r[i]+t[i];
      if i > 1 then M[i] := M[i]+r[i-1]+t[i-1] fi;
      if i < 4 then M[i] := M[i]+r[i+1]+t[i+1] fi;
      if M[i] <> 1 and M[i] <> 4 then return 0 fi;
    fi od;
    1
  end proc:
  T:= Matrix(256,256, q);
```

$$T := \begin{bmatrix} \textit{256 x 256 Matrix} \\ \textit{Data Type: anything} \\ \textit{Storage: rectangular} \\ \textit{Order: Fortran\_order} \end{bmatrix} \qquad \textbf{(1)}$$

Thus $a(n) = u\, T^n\, v$ where u and v are row and column vectors respectively with $u_i = 1$ for i corresponding to configurations with bottom row $(0, 0, 0, 0)$, 0 otherwise, and $v_i = 1$ for i corresponding to configurations with top row $(0, 0, 0, 0)$, 0 otherwise. The following Maple code produces these vectors.

```
> u:= Vector[row](256):
  v:= Vector(256):
```

```
for i from 0 to 15 do u[16*i+1]:= 1; v[i+1]:= 1;
od:
```

To check, here are the first few entries of our sequence.

```
> seq(u . T^n . v, n = 1 .. 10);
```

$$4, 28, 91, 366, 1644, 6545, 26865, 112345, 461363, 1902251 \quad (2)$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \to t^{40} - t^{39} - 3\,t^{38} - 55\,t^{37} - 5\,t^{36} + 28\,t^{35} + 973\,t^{34} + 419\,t^{33} + 241\,t^{32} - 8278\,t^{31} - 4475\,t^{30} - 4251\,t^{29} + 39308\,t^{28} + 21297\,t^{27} + 23721\,t^{26} - 113007\,t^{25} - 54499\,t^{24} - 73183\,t^{23} + 209217\,t^{22} + 80824\,t^{21} + 140425\,t^{20} - 256577\,t^{19} - 66161\,t^{18} - 176558\,t^{17} + 206843\,t^{16} + 18139\,t^{15} + 142548\,t^{14} - 106615\,t^{13} + 14388\,t^{12} - 69376\,t^{11} + 33558\,t^{10} - 12530\,t^{9} + 17736\,t^{8} - 5784\,t^{7} + 2604\,t^{6} - 1792\,t^{5} + 320\,t^{4} \quad (3)$$

This turns out to have degree 40. Thus we will have $0 = u\,P(T)\;T^n\,v = \sum_{i=0}^{40} p_i\,a(i+n)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 40, which would hold true for any u and v. It seems that with our particular u and v we have a recurrence of order only 21, corresponding to a factor of P.

```
> factor(P(t));
```

$$t^4\,(t^6 + 3\,t^4 - 8\,t^3 + 4\,t^2 - 3\,t + 1)\,(t^9 - 8\,t^6 + t^5 + t^4 + 12\,t^3 - 2\,t^2 - 2\,t - 8)\,(t^{21} - t^{20} - 6\,t^{19} - 36\,t^{18} - 8\,t^{17} + 71\,t^{16} + 345\,t^{15} + 221\,t^{14} - 167\,t^{13} - 1167\,t^{12} - 930\,t^{11} - 200\,t^{10} + 1607\,t^{9} + 1327\,t^{8} + 670\,t^{7} - 1036\,t^{6} - 741\,t^{5} - 514\,t^{4} + 306\,t^{3} + 128\,t^{2} + 114\,t - 40) \quad (4)$$

```
> Q:= unapply(t^21-t^20-6*t^19-36*t^18-8*t^17+71*t^16+345*t^15+221*
  t^14-167*t^13-1167*t^12-930*t^11-200*t^10+1607*t^9+1327*t^8+670*
  t^7-1036*t^6-741*t^5-514*t^4+306*t^3+128*t^2+114*t-40, t);
```

$$Q := t \to t^{21} - t^{20} - 6\,t^{19} - 36\,t^{18} - 8\,t^{17} + 71\,t^{16} + 345\,t^{15} + 221\,t^{14} - 167\,t^{13} - 1167\,t^{12} - 930\,t^{11} - 200\,t^{10} + 1607\,t^{9} + 1327\,t^{8} + 670\,t^{7} - 1036\,t^{6} - 741\,t^{5} - 514\,t^{4} + 306\,t^{3} + 128\,t^{2} + 114\,t - 40 \quad (5)$$

The complementary factor $R(t) = \dfrac{P(t)}{Q(t)}$ has degree 19.

```
> R:= unapply(normal(P(t)/Q(t)), t);
```

$$R := t \to (t^{15} + 3\,t^{13} - 16\,t^{12} + 5\,t^{11} - 26\,t^{10} + 80\,t^{9} - 39\,t^{8} + 54\,t^{7} - 117\,t^{6} + 56\,t^{5} - 51\,t^{4} + 74\,t^{3} - 28\,t^{2} + 22\,t - 8)\,t^4 \quad (6)$$

Now we want to show that $b(n) = u\,Q(T)\;T^n\,v = 0$ for all n. This will certainly satisfy the order-19 recurrence

$$\sum_{i=0}^{19} r_i\,b(i+n) = \sum_{i=0}^{19} r_i u\,Q(T)\;T^{n+i} v = u\,Q(T)\;R(T)\;T^n\,v = u\,P(T)\;T^n\,v = 0$$

where r_i are the coefficients of $R(t)$. To show all $b(n) = 0$ it suffices to show $b(0) = ... = b(18) = 0$. This would take some time to do naively, so it's worthwhile to do some pre-calculation.

```
> w:= u . Q(T):
  V[0]:= v:
  for i from 1 to 18 do V[i]:= T . V[i-1] od:
  seq(w . V[i],i=0..18);
```

$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 \tag{7}$$